

---

Subject: Re: Array element deletion

Posted by [Edd Edmondson](#) on Fri, 12 Sep 2003 13:19:03 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Edd Edmondson <eddedmondson@hotmail.com> wrote:

> Attempts using WHERE() have failed for me\* and currently I'm looping  
> through the array of indices and deleting them one at a time - which is  
> fine when I've only got 20 or 30 of them but I doubt it'll scale well.

... and indeed that method is broken, since everytime I delete an element  
the indices of the unwanted ones change....

--

Edd

---

---

Subject: Re: Array element deletion

Posted by [jjbezair](#) on Fri, 12 Sep 2003 13:27:48 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

In article <bjsh0u\$q77\$1@news.ox.ac.uk>,

Edd Edmondson <eddedmondson@hotmail.com> wrote:

> Supposing I have a big array and I also have an array containing indices  
> of unwanted elements. Is there a neat way of removing those elements from  
> the array?

>

> Attempts using WHERE() have failed for me\* and currently I'm looping  
> through the array of indices and deleting them one at a time - which is  
> fine when I've only got 20 or 30 of them but I doubt it'll scale well.

>

>

> \* - I tried to make a list of indices that I wanted to keep using  
> WHERE(indgen(x) ne unwanted) but that only produces the first  
> n\_elements(unwanted) of the correct indices.

>

> --

> Edd

I'm sure one or more of the gurus here will have a faster solution that  
uses histogram, but I've always used index masks for this sort of thing

```
indexMask = lonarr(n_elements(bigArray))
```

```
indexMask(unwanted) = 1
```

```
wanted = where(indexMask == 0)
```

```
smallArray = bigArray(wanted)
```

of course you should check that where returns a nonzero array, etc., and

this method uses 4 times the memory of the original array, but memory is cheap these days, and these 4 lines of code are quick to type...

I'm curious to see what more efficient techniques other people here use.

regards,  
Jeff  
--

---

Subject: Re: Array element deletion  
Posted by [David Fanning](#) on Fri, 12 Sep 2003 13:31:20 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

Edd Edmondson writes:

> Supposing I have a big array and I also have an array containing indices  
> of unwanted elements. Is there a neat way of removing those elements from  
> the array?  
>  
> Attempts using WHERE() have failed for me\* and currently I'm looping  
> through the array of indices and deleting them one at a time - which is  
> fine when I've only got 20 or 30 of them but I doubt it'll scale well.

Here is a function from my web page:

FUNCTION SetDifference, a, b

    ; = a and (not b) = elements in A but not in B

    mina = Min(a, Max=maxa)

    minb = Min(b, Max=maxb)

    IF (minb GT maxa) OR (maxb LT mina) THEN RETURN, a ;No intersection...

    r = Where((Histogram(a, Min=mina, Max=maxa) NE 0) AND \$

        (Histogram(b, Min=mina, Max=maxa) EQ 0), count)

    IF count eq 0 THEN RETURN, -1 ELSE RETURN, r + mina

END

Here is how it works. Suppose you have an array:

    array = [3.5, 4.8, 9.3, 2.1, 7.6, 4.6]

And an array of indices you don't want:

    bad = [2, 4]

You would do this:

```
possible = Indgen(N_Elements(array))
good = SetDifference(possible, bad)
Print, possible
Print, good
newArray = array[good]
Print, newArray
```

You can learn more about these set methods here:

[http://www.dfanning.com/tips/set\\_operations.html](http://www.dfanning.com/tips/set_operations.html)

Cheers,

David

--

David W. Fanning, Ph.D.

Fanning Software Consulting, Inc.

Phone: 970-221-0438, E-mail: [david@dfanning.com](mailto:david@dfanning.com)

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

Toll-Free IDL Book Orders: 1-888-461-0155

---

Subject: Re: Array element deletion

Posted by [Edd Edmondson](#) on Fri, 12 Sep 2003 13:44:19 GMT

[View Forum Message](#) <> [Reply to Message](#)

Excellent, thanks for the speedy response :-)

--

Edd

---

Subject: Re: Array element deletion

Posted by [R.Bauer](#) on Sat, 13 Sep 2003 17:05:20 GMT

[View Forum Message](#) <> [Reply to Message](#)

Edd Edmondson wrote:

```
> Supposing I have a big array and I also have an array containing indices
> of unwanted elements. Is there a neat way of removing those elements from
> the array?
>
> Attempts using WHERE() have failed for me* and currently I'm looping
> through the array of indices and deleting them one at a time - which is
> fine when I've only got 20 or 30 of them but I doubt it'll scale well.
>
```

>  
> \* - I tried to make a list of indices that I wanted to keep using  
> WHERE(indgen(x) ne unwanted) but that only produces the first  
> n\_elements(unwanted) of the correct indices.  
>

Just in addition to Davids answer.

a\_not\_b

[http://www.fz-juelich.de/icg/icg-i/idl\\_icglib/idl\\_source/idl\\_html/dbase/a\\_not\\_b\\_dbase.pro.html](http://www.fz-juelich.de/icg/icg-i/idl_icglib/idl_source/idl_html/dbase/a_not_b_dbase.pro.html)

regards  
Reimar

--

Forschungszentrum Juelich  
email: R.Bauer@fz-juelich.de  
<http://www.fz-juelich.de/icg/icg-i/>

=====

a IDL library at ForschungsZentrum Juelich  
[http://www.fz-juelich.de/icg/icg-i/idl\\_icglib/idl\\_lib\\_intro.html](http://www.fz-juelich.de/icg/icg-i/idl_icglib/idl_lib_intro.html)

---